

# **The PRINTTO Procedure**

Overview 819 Procedure Syntax 820 PROC PRINTTO Statement 820 Concepts 822 Page Numbering 822 Examples 823 Example 1: Routing to External Files 823 Example 2: Routing to SAS Catalog Entries 825 Example 3: Using Procedure Output as an Input File 829

## **Overview**

The PRINTTO procedure defines destinations for SAS procedure output and for the SAS log. By default, SAS procedure output and the SAS log are routed to the default procedure output file and the default SAS log file for your method of operation. See Table 29.1 on page 819. You can store the SAS log or procedure output in an external file or in a SAS catalog entry. With additional programming, you can use SAS output as input data within the same job.

 Table 29.1
 Default Destinations for SAS Log and Procedure Output

Method of running the SAS System	SAS log destination	Procedure output destination
windowing environment	the LOG window	the OUTPUT window
interactive line mode	the display monitor (as statements are entered)	the display monitor (as each step executes)
noninteractive mode or batch mode	depends on the host operating system	depends on the operating environment

Operating Environment Information: For information and examples specific to your operating system or environment, see the appropriate SAS Companion or technical report.  $\triangle$ 

## **Procedure Syntax**

**PROC PRINTTO** <*option(s)*>;

## **PROC PRINTTO Statement**

**Tip:** To reset the destination for the SAS log and procedure output to the default, use the PROC PRINTTO statement without options.

**Tip:** To route the SAS log and procedure output to the same file, specify the same file with both the LOG= and PRINT= options.

#### **PROC PRINTTO** <*option(s)*>;

To do this	Use this option
provide a description for a SAS log or procedure output stored in a SAS catalog entry	LABEL=
route the SAS log to a permanent external file or SAS catalog entry	LOG=
combine the SAS log and procedure output into a single file	LOG= and PRINT= with same destination
replace the file instead of appending to it	NEW=
route procedure output to a permanent external file or SAS catalog entry	PRINT=

### Without Options

Using a PROC PRINTTO statement with no options

- $\hfill\square$  closes any files opened by a PROC PRINTTO statement
- □ points both the SAS log and SAS procedure output to their default destinations.

**Interaction:** To close the appropriate file and to return only the SAS log or procedure output to its default destination, use LOG=LOG or PRINT=PRINT.

Featured in: Example 1 on page 823 and Example 2 on page 825

### **Options**

#### LABEL='*description*'

provides a description for a catalog entry that contains a SAS log or procedure output. **Range:** 1 to 40 characters

**Interaction:** Use the LABEL= option only when you specify a catalog entry as the value for the LOG= or the PRINT= option.

#### Featured in: Example 2 on page 825

#### LOG=LOG | file-specification | SAS-catalog-entry

routes the SAS log to one of three locations:

#### LOG

routes the SAS log to its default destination.

#### file-specification

routes the SAS log to an external file. It is one of the following:

#### 'external-file'

the name of an external file specified in quotation marks.

#### fileref

a fileref previously assigned to an external file.

#### SAS-catalog-entry

routes the SAS log to a SAS catalog entry. By default, *libref* is SASUSER, *catalog* is PROFILE, and *type* is LOG. Express *SAS-catalog-entry* in one of the following ways:

#### libref.catalog.entry<.LOG>

a SAS catalog entry stored in the SAS data library and SAS catalog specified.

#### catalog.entry<.LOG>

a SAS catalog entry stored in the specified SAS catalog in the default SAS data library SASUSER.

#### entry.LOG

a SAS catalog entry stored in the default SAS library and catalog: SASUSER.PROFILE.

### Default: LOG.

- **Tip:** After routing the log to an external file or a catalog entry, you can specify LOG to route the SAS log back to its default destination.
- **Tip:** When routing the SAS log, include a RUN statement in the PROC PRINTTO statement. If you omit the RUN statement, the first line of the following DATA or PROC step is not routed to the new file. (This occurs because a statement does not execute until a step boundary is crossed.)
- **Interaction:** The NEW option replaces the existing contents of a file with the new log. Otherwise, the new log is appended to the file.
- **Interaction:** To route the SAS log and procedure output to the same file, specify the same file with both the LOG= and PRINT= options.
- **Interaction:** When routing the log to a SAS catalog entry, you can use the LABEL option to provide a description for the entry in the catalog directory.
- Featured in: Example 1 on page 823, Example 2 on page 825, and Example 3 on page 829

#### NEW

clears any information that exists in a file and prepares the file to receive the SAS log or procedure output.

- **Default:** If you omit NEW, the new information is appended to the existing file.
- Interaction: If you specify both LOG= and PRINT=, NEW applies to both.
- **Featured in:** Example 1 on page 823, Example 2 on page 825, and Example 3 on page 829
- **PRINT= PRINT | file-specification | SAS-catalog-entry**

routes procedure output to one of three locations:

#### PRINT

routes procedure output to its default destination. After routing it to an external file or a catalog entry, you can specify PRINT to route subsequent procedure output to its default destination.

#### file-specification

routes procedure output to an external file. It is one of the following:

'external-file'

the name of an external file specified in quotation marks.

fileref

a fileref previously assigned to an external file.

SAS-catalog-entry

routes procedure output to a SAS catalog entry. By default, *libref* is SASUSER, *catalog* is PROFILE, and *type* is LOG. Express *SAS-catalog-entry* in one of the following ways:

#### *libref.catalog.entry*<.LOG>

a SAS catalog entry stored in the SAS data library and SAS catalog specified.

#### catalog.entry<.LOG>

a SAS catalog entry stored in the specified SAS catalog in the default SAS data library SASUSER.

#### entry.LOG

a SAS catalog entry stored in the default SAS library and catalog: SASUSER.PROFILE.

### Aliases: FILE=, NAME=

#### Default: PRINT

**Interaction:** The NEW option replaces the existing contents of a file with the new procedure output. If you omit NEW, the new output is appended to the file.

**Interaction:** To route the SAS log and procedure output to the same file, specify the same file with both the LOG= and PRINT= options.

**Interaction:** When routing procedure output to a SAS catalog entry, you can use the LABEL option to provide a description for the entry in the catalog directory.

Featured in: Example 3 on page 829

#### UNIT=nn

routes the output to the file identified by the fileref FTnnF001, where nn is an integer between 1 and 99.

**Range:** 1 to 99, integer only.

**Tip:** You can define this fileref yourself; however, some operating systems predefine certain filerefs in this form.

## **Concepts**

## Page Numbering

□ When the SAS system option NUMBER is in effect, there is a single page-numbering sequence for all output in the current job or session. When NONUMBER is in effect, output pages are not numbered.

□ You can specify the beginning page number for the output you are currently producing by using the PAGENO= in an OPTIONS statement.

## **Examples**

## **Example 1: Routing to External Files**

Procedure features: PRINTTO statement: Without options Options: LOG= NEW PRINT=

This example uses PROC PRINTTO to route the log and procedure output to an external file and then reset both destinations to the default.

### Program

SOURCE writes the lines of source code to the default destination for the SAS log. See Output 29.1 on page 824.

options nodate pageno=1 linesize=80 pagesize=60 source;

PROC PRINTTO uses LOG= to route the SAS log to an external file. By default, this log is appended to the current contents of **log-file**. See Output 29.2 on page 824.

```
proc printto log='log-file';
    run;
```

The DATA step creates the NUMBERS data set.

```
data numbers;
input x y z;
datalines;
14.2 25.2 96.8
10.8 51.6 96.8
9.5 34.2 138.2
8.8 27.6 83.2
```

11.5 49.4 287.0 6.3 42.0 170.7

PROC PRINTTO routes output to an external file. Because NEW is specified, any output written to **output-file** will overwrite the file's current contents. The PROC PRINT output is written to the specified external file. See Output 29.3 on page 825.

```
proc printto print='output-file' new;
run;
```

```
proc print data=numbers;
    title 'Listing of NUMBERS Data Set';
run;
```

PROC PRINTTO routes subsequent logs and procedure output to their default destinations and closes both of the current files. See Output 29.1 on page 824.

```
proc printto;
run;
```

### Log

Output 29.1 Portion of Log Routed to the Default Destination

```
1 options nodate pageno=1 linesize=80 pagesize=60 source;
2 proc printto log='log-file';
3 run;
```

Output 29.2 Portion of Log Routed to an External File

```
5
6
          data numbers;
7
             input x y z;
8
             datalines;
NOTE: The data set WORK.NUMBERS has 6 observations and 3 variables.
NOTE: DATA statement used:
     real time
                        0.00 seconds
     cpu time
                        0.00 seconds
15
          ;
          proc printto print='output-file' new;
16
16
17
          run;
NOTE: PROCEDURE PRINTTO used:
                       0.00 seconds
     real time
     cpu time
                        0.00 seconds
18
19
          proc print data=numbers;
20
            title 'Listing of NUMBERS Data Set';
21
          run;
NOTE: The PROCEDURE PRINT printed page 1.
NOTE: PROCEDURE PRINT used:
                  0.00 seconds
     real time
                        0.00 seconds
     cpu time
22
23
          proc printto;
24
          run;
```

## Output

Output 29.3 Procedure Output Routed to an External File

Listing of NUMBERS Data Set					1
OBS	x	У	z		
1	14.2	25.2	96.8		
2	10.8	51.6	96.8		
3	9.5	34.2	138.2		
4	8.8	27.6	83.2		
5	11.5	49.4	287.0		
6	6.3	42.0	170.7		

## **Example 2: Routing to SAS Catalog Entries**

**Procedure features:** 

PRINTTO statement: Without options Options: LABEL= LOG= NEW PRINT=

This example uses PROC PRINTTO to route the SAS log and procedure output to a SAS catalog entry and then to reset both destinations to the default.

### Program

```
libname lib1 'SAS-data-library';
options nodate pageno=1 linesize=80 pagesize=60 source;
```

PROC PRINTTO routes the SAS log to a SAS catalog entry named SASUSER.PROFILE.TEST.LOG. The PRINTTO procedure uses the default libref and catalog SASUSER.PROFILE because only the entry name and type are specified. LABEL= assigns a description for the catalog entry. See Output 29.4 on page 828.

```
proc printto log=test.log label='Inventory program' new;
run;
```

The DATA step creates a permanent SAS data set.

```
data lib1.inventry;
length Dept $ 4 Item $ 6 Season $ 6 Year 4;
input dept item season year @@;
datalines;
3070 20410 spring 1996 3070 20411 spring 1997
3070 20412 spring 1997 3070 20413 spring 1997
3070 20414 spring 1996 3070 20416 spring 1995
3071 20500 spring 1994 3071 20501 spring 1995
3071 20502 spring 1996 3071 20503 spring 1996
3071 20505 spring 1994 3071 20506 spring 1994
3071 20507 spring 1994 3071 20424 spring 1994
;
```

PROC PRINTTO routes procedure output from the subsequent PROC REPORT to the SAS catalog entry LIB1.CAT1.INVENTRY.OUTPUT. LABEL= assigns a description for the catalog entry. See Output 29.5 on page 828.

PROC PRINTTO closes the current files opened by PROC PRINTTO and reroutes subsequent SAS logs and procedure output to their default destinations.

proc printto; run;

### Log

**Output 29.4** SAS Log Routed to SAS Catalog Entry SASUSER.PROFILE.TEST.LOG.

You can view this catalog entry in the BUILD window of the SAS Explorer.

```
8
9
    data lib1.inventry;
10
       length Dept $ 4 Item $ 6 Season $ 6 Year 4;
       input dept item season year 00;
11
12
       datalines;
NOTE: SAS went to a new line when INPUT statement reached past the end of a
     line.
NOTE: The data set LIB1.INVENTRY has 14 observations and 4 variables.
NOTE: DATA statement used:
     real time 0.00 seconds
                        0.00 seconds
     cpu time
20
    ;
21
    proc printto print=lib1.cat1.inventry.output
22
23
      label='Inventory program' new;
24
    run;
NOTE: PROCEDURE PRINTTO used:
     real time 0.00 seconds
     cpu time
                        0.00 seconds
25
26
    proc report data=lib1.inventry nowindows headskip;
27
       column dept item season year;
       title 'Current Inventory Listing';
28
   run;
29
NOTE: PROCEDURE REPORT used:
     real time
                        0.00 seconds
                       0.00 seconds
     cpu time
30
31
    proc printto;
32
    run;
```

## Output

Output 29.5 Procedure Output Routed to SAS Catalog Entry LIB1.CAT1.INVENTRY.OUTPUT.

Current Inventory Listing 1 Dept Item Season Year 3070 20410 spring 1996 3070 20411 spring 1997 3070 20412 1997 spring 3070 20413 spring 1997 3070 20414 1996 spring 3070 20416 1995 spring 3071 20500 1994 spring 3071 20501 spring 1995 3071 20502 spring 1996 3071 20503 1996 spring 3071 20505 spring 1994 3071 20506 1994 spring 3071 20507 1994 spring 3071 20424 spring 1994

You can view this catalog entry in the BUILD window of the SAS Explorer.

## **Example 3: Using Procedure Output as an Input File**

#### Procedure features:

PRINTTO statement: Without options Options: LOG= NEW PRINT=

This example uses PROC PRINTTO to route procedure output to an external file and then uses that file as input to a DATA step.

The DATA step uses the RANUNI function to randomly generate values for variables X and Y in data set A.

```
data test;
    do n=1 to 1000;
        x=int(ranuni(77777)*7);
        y=int(ranuni(77777)*5);
        output;
        end;
run;
```

The FILENAME statement assigns a fileref to an external file. PROC PRINTTO routes subsequent procedure output to the file referenced by the fileref ROUTED. See Output 29.6 on page 830.

```
filename routed 'output-filename';
proc printto print=routed new;
run;
```

PROC FREQ computes frequency counts and a chi-square analysis on variables X and Y in data set TEST. This output is routed to the file referenced as ROUTED.

```
proc freq data=test;
    tables x*y / chisq;
run;
```

You must use another PROC PRINTTO to close the file referenced by fileref ROUTED so that the following DATA step can read it. The step also routes subsequent procedure output to the default destination. PRINT= causes the step to affect only procedure output, not the SAS log.

```
proc printto print=print;
run;
```

The DATA step uses ROUTED, the file containing PROC FREQ output, as an input file and creates data set PROBTEST. This DATA step reads all records in ROUTED but creates an observation only from a record that begins with **Chi-Squa**.

```
data probtest;
infile routed;
input wordl $ @;
if wordl='Chi-Squa' then
    do;
        input df chisq prob;
        keep chisq prob;
        output;
        end;
run;
```

PROC PRINT produces a simple listing of data set PROBTEST. This output is routed to the default destination. See Output 29.7 on page 831.

```
proc print data=probtest;
    title 'Chi-Square Analysis for Table of X by Y';
run;
```

		The FF	REQ Proced	lure		
		Table	e of x by	У		
x	У					
Frequency Percent Row Pct Col Pct	0	1	2	3	4	Total
0	29 2.90 23.02 15.18	33   3.30   26.19   16.18	12   1.20   9.52   6.25	25   2.50   19.84   11.74	27   2.70   21.43   13.50	126 12.60
1	23 2.30 19.66 12.04	26 2.60 22.22 12.75	29 2.90 24.79 15.10	20 2.00 17.09 9.39	19   1.90   16.24   9.50	117 11.70
2	28 2.80 19.86 14.66	26 2.60 18.44 12.75	32   3.20   22.70   16.67	30 3.00 21.28 14.08	25 2.50 17.73 12.50	141 14.10
3	26 2.60 15.95 13.61	24 2.40 14.72 11.76	36 3.60 22.09 18.75	32 3.20 19.63 15.02	45 4.50 27.61 22.50	163 16.30
4	25 2.50 16.78 13.09	31 3.10 20.81 15.20	28 2.80 18.79 14.58	36 3.60 24.16 16.90	29   2.90   19.46   14.50	149 14.90
5	32 3.20 21.77 16.75	29 2.90 19.73 14.22	26 2.60 17.69 13.54	33   3.30   22.45   15.49	27   2.70   18.37   13.50	147 14.70
6	28 2.80 17.83 14.66	35 3.50 22.29 17.16	29 2.90 18.47 15.10	37   3.70   23.57   17.37	28 2.80 17.83 14.00	157 15.70
Total	191 19.10	204 20.40	192 19.20	213 21.30	200 20.00	1000 100.00
		The FF	REQ Proced	lure		
	Sta	tistics f	or Table	of x by y		
Statis	Stic		DF	Val	ue F 	rob
Chi-So Likel: Mante Phi Co Contin Crame	quare ihood Rati L-Haenszel Defficient Igency Coe c's V	lare         24         27.2971         0.2           lood Ratio Chi-Square         24         28.1830         0.2           Haenszel Chi-Square         1         0.6149         0.4           officient         0.1652         0.1630         0.826				908 524 330
		Sample	e Size = 1	.000		

### Output 29.6 PROC FREQ Output Routed to the External File Referenced as ROUTED

CI	3				
	Obs	chisq	prob		
	1	27.297	0.291		

Output 29.7 PROC PRINT Output of Data Set PROBTEST, Routed to Default Destination

The correct bibliographic citation for this manual is as follows: SAS Institute Inc., SAS<sup>®</sup> Procedures Guide, Version 8, Cary, NC: SAS Institute Inc., 1999. 1729 pp.

#### SAS<sup>®</sup> Procedures Guide, Version 8

Copyright © 1999 by SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-482-9

All rights reserved. Printed in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

**U.S. Government Restricted Rights Notice.** Use, duplication, or disclosure of the software and related documentation by the U.S. government is subject to the Agreement with SAS Institute and the restrictions set forth in FAR 52.227–19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

 $SAS^{\circledast}$  and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries.<sup>®</sup> indicates USA registration.

 $\rm IBM^{\circledast}$  and  $\rm DB2^{\circledast}$  are registered trademarks or trademarks of International Business Machines Corporation. ORACLE  $^{\circledast}$  is a registered trademark of Oracle Corporation.  $^{\circledast}$  indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.